# A Brief Incomplete Introduction to Prover9

This introduction simplifies many aspects of Prover9, focusing on one way of using Prover9 to perform logical inference. Much more information can be found in the Prover9 manual, which is at `http://www.cs.unm.edu/~mccune/mace4/manual/Aug-2007/`, and the official Prover9 web site `http://www.cs.unm.edu/~mccune/mace4/`, where you can download the code.

Prover9 is a resolution/paramodulation automated theorem prover for first-order and equational logic. Prover9 is a successor of the Otter prover, which I have used in previous AI classes. Prover9 implements the resolution inference rule and unit resolution inference rule (respectively called "binary resolution" and "unit deletion" in Prover9). Prover9 also implements other inference rules that we will not discuss.

To use Prover9, one must specify the parameters of its inference procedure, the sentence to be proved, and the sentences that can be used in the proof. All of these are put in an input file. If `file.in` is the input file, one can then call Prover9 by:

```
setenv PROVER9 /home/bylander/prover9/linux/bin/prover9
or PROVER9=/home/bylander/prover9/linux/bin/prover9
$PROVER9 < file.in
```

## Parameters

The parameters of the inference procedure are specified at the beginning of the program. For this class, I suggest that you always use the following:

```
set(binary_resolution).
clear(print_initial_clauses).
clear(print_kept).
clear(print_given).
assign(max_seconds,1).
assign(stats,none).
```

Note that every line ends with a period. Commenting out any of the `clear` lines, as well as changing the `stats` level, will produce more output. A comment starts with a `%` and goes to the end of the line.

## Propositional Logic Example

This example show how Prover9 performs the inference of $W_{1,3}$ in the wumpus world from the book, using only the resolution inference rule.

### Sos List

The sentences to be used are put the tt sos list of formulas. The first three lines list the perceptions. The next three lines represent implications from the lack of a stench. The last line represents the implication from a stench.

```
formulas(sos).
-S11.   -B11.
-S21.   B21.
S12.    -B12.
S11 | -W11.     S11 | -W12.     S11 | -W21.
S21 | -W11.     S21 | -W21.     S21 | -W22.     S21 | -W31.
S12 | -W11.     S12 | -W12.     S12 | -W22.     S12 | -W13.
-S12 | W11 | W12 | W22 | W13.
end_of_list.
```

The `|` operator corresponds to OR, and the `-` operator corresponds to negation. `S11`, `B11`, `S21`, `B21`, `S12`, `B12`, `W11`, `W12`, `W21`, `W22`, `W31`, and `W13` are propositional variables. Note that each clause ends with a period.

### Goal List

The statement to be proved is put into the goal list.

```
formulas(goals).
W13.
end_of_list.
```

In this example, we want to prove that the wumpus is in square $[1, 3]$. Eventually, Prover9 will refute (or deny) this sentence as part of its proof.

### Result

When the parameters, the usable list, and the sos list are all put into a file that is input to Prover9, then the output includes the following proof:

```
1 W13 # label(non_clause) # label(goal).  [goal].
2 -S11.  [assumption].
4 -S21.  [assumption].
6 S12.  [assumption].
8 S11 | -W11.  [assumption].
9 -W11.  [copy(8),unit_del(a,2)].
10 S11 | -W12.  [assumption].
11 -W12.  [copy(10),unit_del(a,2)].
16 S21 | -W22.  [assumption].
17 -W22.  [copy(16),unit_del(a,4)].
20 -S12 | W11 | W12 | W22 | W13.  [assumption].
21 W13.  [copy(20),unit_del(a,6),unit_del(b,9),unit_del(c,11),unit_del(d,17)].
22 -W13.  [deny(1)].
23 $F.  [copy(22),unit_del(a,21)].
```

Each sentence that is given or inferred is numbered. Some numbers are skipped because they correspond to other sentences that were given or inferred, but were not part of the proof.

Line 1 is the goal. Lines 2, 4, 6, 8, 10, and 16 are sentences from the sos list. Lines 9, 11, and 17 show unit resolution inferences: respectively, using sentences 8 and 2, using sentences 10 and 2, and using sentences 16 and 4. Line 21 shows multiple unit resolution inferences to eliminate all `W13` from sentence 20. Line 22 is the negation of sentence 1, which we want to prove. Finally, line 23 indicates a contradiction between sentences 22 and 22, so that we can conclude that the wumpus is in $[1, 3]$.

## Predicate Logic Example

This example infers an ancestor relationship from knowledge about who is a mother/father of whom.

### Sos List

The sentences to be used are put in the usable list.

```
formulas(sos).
mother(Liz,Charley).
father(Charley,Billy).
-mother(x,y) | parent(x,y).
-father(x,y) | parent(x,y).
-parent(x,y) | ancestor(x,y).
-parent(x,y) | -ancestor(y,z) | ancestor(x,z).
end_of_list.
```

The `|` operator corresponds to OR, and the `-` operator corresponds to negation. `mother`, and `father`, `parent`, and `ancestor`, are predicates. `x` and `y` are variables. `Liz`, `Charley`, and `Billy` are constants. The way that Prover9 distinguishes variables from constants is that variables start with a lower-case `u`, `v`, `w`, `x`, `y`, or `z`.

The six sentences are intended to have the following meanings:

Liz is the mother of Charley.
Charley is the father of Billy,
If $x$ is the mother of $y$, then $x$ is a parent of $y$.
If $x$ is the father of $y$, then $x$ is a parent of $y$.
If $x$ is a parent of $y$, then $x$ is an ancestor of $y$.
If $x$ is a parent of $y$ and $y$ is an ancestor of $z$, then $x$ is an ancestor of $z$.

These statements can be represented in first-order predicate logic by:

$\forall x, y\,(\mathit{parent}(x, y) \rightarrow \mathit{ancestor}(x, y))$
$\forall x, y\,((\mathit{parent}(x, y) \wedge \mathit{ancestor}(y, z)) \rightarrow \mathit{ancestor}(x, z))$
$\forall x, y\,(\mathit{mother}(x, y) \rightarrow \mathit{parent}(x, y))$
$\forall x, y\,(\mathit{father}(x, y) \rightarrow \mathit{parent}(x, y))$
$\mathit{mother}(\mathit{Liz}, \mathit{Charley})$
$\mathit{father}(\mathit{Charley}, \mathit{Billy})$

The first-order sentences are converted to clause form by removing universal quantifiers, and transforming to disjunctions.

### Goal List

The sentence to be proved is put into the goal list.

```
list(goals).
ancestor(Liz, Billy).
end_of_list.
```

In this example, we want to prove that Liz is an ancestor of Billy. Eventually, Prover9 will refute (or deny) this sentence as part of its proof.

### Result

When the parameters, the usable list, and the sos list are all put into a file that is input to Prover9, then the output includes the following proof:

```
1 ancestor(Liz,Billy) # label(non_clause) # label(goal).  [goal].
2 -mother(x,y) | parent(x,y).  [assumption].
3 -parent(x,y) | -ancestor(y,z) | ancestor(x,z).  [assumption].
4 -parent(x,y) | ancestor(x,y).  [assumption].
5 -father(x,y) | parent(x,y).  [assumption].
6 -mother(x,y) | -ancestor(y,z) | ancestor(x,z).  [resolve(2,b,3,a)].
7 mother(Liz,Charley).  [assumption].
10 father(Charley,Billy).  [assumption].
11 -father(x,y) | ancestor(x,y).  [resolve(5,b,4,a)].
12 -ancestor(Liz,Billy).  [deny(1)].
13 -ancestor(Charley,x) | ancestor(Liz,x).  [resolve(6,a,7,a)].
16 ancestor(Charley,Billy).  [resolve(11,a,10,a)].
17 -ancestor(Charley,Billy).  [resolve(13,b,12,a)].
18 $F.  [resolve(17,a,16,a)].
```

Each sentence that is given or inferred is numbered. Some numbers are skipped because they correspond to other sentences that were given or inferred, but were not part of the proof. Line 1 is the goal. Lines 2–5, 7, and 10 are sentences given in the sos list. Line 6 shows that (binary) resolution was applied to sentences 2 and 3 to infer that, for all $x, y, z$, if $x$ is the mother of $y$ and $y$ is an ancestor of $z$, then $x$ is an ancestor of $z$. Line 11 shows that binary resolution on sentences 5 and 4 infers, for all $x, y$, if $x$ is the father of $y$, then $x$ is an ancestor of $y$. Line 12 is the negation of the goal. Line 13 shows that binary resolution on sentences 6 and 7 infers, for all $x$, if Charley is an ancestor of $x$, then Liz is an ancestor of $x$. Line 16 shows that binary resolution on sentences 11 and 10 infers that Charley is an ancestor of Billy. Line 17 shows that binary resolution on sentences 13 and 12 infers that Charley is not an ancestor of Billy. Line 18 indicates a contradiction between sentences 17 and 16, so we can conclude that Liz is an ancestor of Billy.